COMPUTER GRAFICA APPLICATA: TRIGONOMETRIA

Seconda parte

Il triangolo è il poligono con il minor numero di lati, è quindi la figura geometrica più elementare e più facilmente comprensibile a tutti. È un poligono importante al punto che una intera parte della matematica e della geometria gli è dedicata: la trigonometria. Infatti, poichè ogni figura piana è rappresentabile come insieme di più triangoli (collegando i vari vertici con segmenti), con la trigonometria si può risolvere gran parte dei problemi di geometria.

Fortunatamente tutti i microcomputer che utilizzano il BASIC interprete, anche non molto esteso, "conoscono" la trigonometria, hanno infatti nel proprio set di istruzioni, le istruzioni trigonometriche classiche, mentre quelle meno classiche sono immediatamente derivabili dalle prime.

Noi, dunque, in questo e nel prossimo articolo studieremo un po' di trigonometria, ovviamente vedendola in relazione soprattutto alla computer grafica.

La risoluzione dei triangoli

Il triangolo è costituito da sei elementi misurabili: tre lati e tre angoli. Quando di un triangolo sono noti tre elementi, fra cui almeno un lato, essendo tutti gli altri calcolabili esso è definito completamente.

Affronteremo l'argomento risoluzione dei triangoli, un classico dei calcolatori tascabili programmabili, e ovviamente affrontabili anche con i micro, in uno dei prossimi articoli. Ora ci soffermeremo invece sulle funzioni trigonometriche, che sono alla base di tutta la trigonometria, cercando dapprima di farle comprendere anche a chi non si ricorda nulla degli studi liceali e poi di utilizzarle in facili programmi grafici applicativi.

Le funzioni trigonometriche

Le funzioni trigonometriche sono valori caratteristici propri di ciascun angolo, e sono alla base di tutti i calcoli trigonometrici.

Osservando il cerchio trigonometrico (figura 1), cerchio di raggio pari alla unità, si possono con facilità comprendere i significati di ciascuna delle sei funzioni trigonometriche elementari: seno, coseno, tangente, secante, cosecante, cotangente.

Il cerchio trigonometrico si suddivide in quattro porzioni dette quadranti e ciascuna funzione assume in ciascun quadrante valori caratteristici.

Nei libri di trigonometria sono riportate varie tabelle sia per la determinazione dei valori caratteristici assunti dalle funzioni trigonometriche per vari angoli caratteristici, sia tabelle che riportano le relazioni esistenti tra le varie funzioni trigonometriche di uno stesso angolo (ovvero ciascuna funzione può essere espressa tramite una espressione contenente ciascuna altra funzione).

Esistono poi tabelle che mettono in relazione funzioni trigonometriche di angoli complementari (ovvero A = 90° - B), tabelle che mettono in relazione funzioni trigonometriche di angoli diversi (es. SIN (A+B) = ...), oppure tabelle di relazione tra le funzioni trigonometriche di multipli e sottomultipli di un determinato angolo (es. SIN (2*A) = ...), ecc. ecc.

Esistono poi leggi trigonometriche relative ai triangoli che mettono in relazione gli angoli e i lati di ciascun triangolo.

Non vogliamo ovviamente parlare di tutto questo, prenderemo solo quello che ci serve per i nostri programmi, cercando come al solito di spingere i lettori più interessati allo studio di un buon testo di trigonometria.

I valori delle funzioni trigonometriche

Il programma Funzioni Trigonometriche (listing in figura 2 e output su carta in figura 3) crea una tabella nella quale per ciascun angolo caratteristico vengono calcolati i corrispondenti valori delle funzioni trigonometriche.



Figura 1 - IL CERCHIO TRIGONOMETRICO. Poichè il suo raggio è pari ad uno, i vari segmenti individuati danno il valore (grandezza adimensionale) delle varie funzioni trigonometriche.

Il BASIC del microcomputer conosce solo gli angoli espressi in radianti (frazioni o multipli di PIgreco) occorre quindi, ogni volta, per la traduzione in gradi, eseguire la proporzione PIgreco: Arad. = 180°: Agradi.

Non tutte le funzioni sono presenti nel linguaggio e allora si ricorre a formule di traduzione. Ad esempio nel programma in esame sono calcolati direttamente seno, coseno, e tangente, invece sono espresse con formula secante, cosecante e cotangente.

Del programma precedente abbiamo realizzato anche una versione grafica, con uscita su monitor HGR dell'Apple II (listing in figura 4 e output in figura 5).

Il programma tramite un loop ripetuto sei volte visualizza le curve corrispondenti ai valori assunti delle sei funzioni al variare dell'angolo da - PIgreco a + PIgreco. Nella riga 120 va precisato il fattore di scala, ovvero il fattore per cui si vuol moltiplicare il dato trovato per visualizzarlo adeguatamente sul monitor. In pratica tale valore corrisponde al valore della unità in direzione Y, e quindi sono state tracciate le rette Y = 1 nella scala prescelta.

Il valore 44.44 in riga 410 è invece il fattore di scala nella direzione X, che essendo l'intervallo della X da - PIgreco a + PIgreco ed essendo la definizione dell'Apple II di 280 pixel in direzione X sarà 280/6.28.

```
120  PR# 3: PRINT "I/C100N": HOME :PI = 3.1415926:X1 = .00001
130  PRINT "TABELLA VALORI CARATTERISTICI"
140  PRINT "DELLE FUNZIONI TRIGONOMETRICHE"
150  PRINT "  ANG.RAD.ANG.GRD  SENO    COSENO";
160  PRINT " TANGENTE COT-NTE SECANTE COS-NTE"
170  FOR X = 0 TO 2 * PI + .001 STEP PI / 12
180 R = X:A = 180 * X / PI
190 K = X: GOSUB 400:K = A: GOSUB 400
200 K =  SIN (X): GOSUB 400: REM   SENO
210 K =  COS (X): GOSUB 400: REM               COSENO
220 K =  TAN (X): GOSUB 400: REM  TANGENTE
230 K = 1 /  TAN (X + X1): GOSUB 400: REM  COTANGENTE
240 K = 1 /  COS (X + X1): GOSUB 400: REM  SECANTE
250 K = 1 /  SIN (X + X1): GOSUB 400: REM  COSECANTE
260  PRINT : NEXT X: END
290  PRINT : NEXT X: END
400  REM  ARROTONDAMENTO
410 K =  INT ((K + .0005) * 1000) / 1000
420 K$ =  STR$ (K)
430  IF  ABS (K) > 999 THEN K$ = "INF"
440  IF  ABS (K) < .0001 THEN K$ = "0.0"
450  PRINT  RIGHT$ ("          " + K$,8);: RETURN
```

Figura 2 - PROGRAMMA FUNZIONI TRIGONOMETRICHE - Listing. Il computer conosce solo gli angoli espressi in radianti, per ottenere il valore in gradi si utilizza la solita proporzione Agradi: Arad. = 180: 3,14. N.B.: I/C = CTRL-I nella linea 120.

TABELLA VALORI CARATTERISTICI
DELLE FUNZIONI TRIGONOMETRICHE

ANG. RAD.	ANG. GRD	SENO	COSENO	TANGENTE	COT-NTE	SECANTE	COS-NTE
0.0	0.0	0.0	1	0.0	INF	1	INF
.262	15	.259	.966	.268	3.732	1.035	3.864
.524	30	.5	.866	.577	1.732	1.155	2
.785	45	.707	.707	1	1	1.414	1.414
1.047	60	.866	.5	1.732	.577	2	1.155
1.309	75	.966	.259	3.732	.268	3.864	1.035
1.571	90	1	0.0	INF	0.0	INF	1
1.833	105	.966	-.259	-3.732	-.268	-3.864	1.035
2.094	120	.866	-.5	-1.732	-.577	-2	1.155
2.356	135	.707	-.707	-1	-1	-1.414	1.414
2.618	150	.5	-.866	-.577	-1.732	-1.155	2
2.88	165	.259	-.966	-.268	-3.732	-1.035	3.864
3.142	180	0.0	-1	0.0	INF	-1	INF
3.403	195	-.259	-.966	.268	3.732	-1.035	-3.864
3.665	210	-.5	-.866	.577	1.732	-1.155	-2
3.927	225	-.707	-.707	1	1	-1.414	-1.414
4.189	240	-.866	-.5	1.732	.577	-2	-1.155
4.451	255	-.966	-.259	3.732	.268	-3.864	-1.035
4.712	270	-1	0.0	INF	0.0	INF	-1
4.974	285	-.966	.259	-3.732	-.268	3.864	-1.035
5.236	300	-.866	.5	-1.732	-.577	2	-1.155
5.498	315	-.707	.707	-1	-1	1.414	-1.414
5.76	330	-.5	.866	-.577	-1.732	1.155	-2
6.021	345	-.259	.966	-.268	-3.732	1.035	-3.864
6.283	360	0.0	1	0.0	INF	1	INF

Figura 3 - PROGRAMMA FUNZIONI TRIGONOMETRICHE - Output. Per la realizzazione della tabella è stata introdotta una subroutine di arrotondamento, anche per coprire i casi in cui il computer darebbe errore divisione per zero.

Il programma sinusoide

Se osserviamo il cerchio trigonometrico notiamo che incrementando l'angolo X di un angolo giro i valori delle funzioni trigonometriche ritornano uguali a quelli di partenza.

Ovvero la funzione Y = SIN (X) non è biunivoca, cioè mentre ad un valore di X corrisponde un solo valore di Y non è vero il viceversa, ad un Y corrispondono infiniti valori di X separati gli uni dagli altri di un angolo giro.

Generalizzando la formula Y = A + B* SIN (X) e inserendola in un loop con il quale variano i valori A e B abbiamo realizzato il programma listato in figura 6 e il cui output è in figura 7.

Il loop di calcolo della X è compreso tra -3* PIgreco e + 3* PIgreco e il suo step è il valore SX in riga 20.

Questo, come altri programmi presentati, ha una uscita su monitor Apple oppure su Plotter, a seconda che il flag FL sia uguale a 1 o uguale a O.

Il programma cicloide

Le curve in cui appaiono le funzioni trigonometriche sono innumerevoli. Anzi non è neanche necessario andarle a cercare sui libri, basta un po' di fantasia per inventarne tantissime. Basta fare Y = ... e poi mettere un po' di seni, coseni, tangenti con un po' di altre operazioni matematiche e poi vedere cosa succede.

La cicloide è invece una curva classica. Rappresenta il moto di un punto solidale con un cerchio che rotola su di un piano. Se il punto è coincidente con il centro del cerchio il moto è una retta, se il punto è sul bordo del cerchio il moto sarà ciclico toccando con il minimo la base di rotolamento e con il massimo una altezza pari a 2*R sulla base.

Se poi il punto dista più di R dal centro del cerchio il punto minimo sarà al di sotto della base.

Il programma (listato in figura 8) esamina il moto per un rotolamento da -2.5* PIgreco a +2.5* PIgreco e traccia dieci curve per punti distanti dal centro da 0*R a 3*R, con un intervallo tra di loro di R/3.

Abbiamo realizzato due output (figura 9 e figura 10) per far vedere come la semplice sostituzione dello STEP da PI/20 a PI/2 comporta output diversissimi tra di loro.

Coordinate cartesiane e coordinate polari

In molte applicazioni trigonometriche si utilizza la rappresentazione polare, nella quale il sistema di riferimento è costituito da un'origine O e da un'asse di riferimento

```
100  FOR J = 1 TO 6: READ A$(J): NEXT
110  DATA   SENO, COSENO, TANGENTE, COTANGENTE, SECANTE, COSECANTE
120  TEXT : HOME :PI = 3.1415926:FS = 40: HGR : HCOLOR= 3
130  HPLOT 0,0 TO 279,0 TO 279,159 TO 0,159 TO 0,0
140  HPLOT 0,80 TO 279,80: HPLOT 140,0 TO 140,159
150  HPLOT 0,80 - FS TO 279,80 - FS: HPLOT 0,80 + FS TO 279,80 + FS
160  HPLOT 70,0 TO 70,159: HPLOT 210,0 TO 210,159
170  FOR J = 1 TO 6: GOSUB 500: FOR X =  - PI TO PI STEP PI / 70
180  ON J GOTO 200,210,220,230,240,250
200 K =  SIN (X): GOTO 260
210 K =  COS (X): GOTO 260
220 K =  TAN (X): GOTO 260
230 K = 1 /  TAN (X): GOTO 260
240 K = 1 /  COS (X): GOTO 260
250 K = 1 /  SIN (X): GOTO 260
260  GOSUB 400: NEXT X,J: END
400  REM  PLOTTAGGIO PUNTO
410 XS = X * 44.4 + 140.5:YS = 80.5 - K * FS
420  IF YS > 159 THEN YS = 159
430  IF YS < 0 THEN YS = 0
440  HPLOT XS,YS: RETURN
500  REM  SCRITTURA FONDO PAGINA
510  HOME
520  VTAB (21): PRINT "-PI     -PI/2         0         PI/2        PI"
530  VTAB (23): HTAB (28): PRINT "FATT.SCALA ";FS;
540  VTAB (23): PRINT A$(J);: RETURN
```

Figura 4 - PROGRAMMA GRAFICO FUNZIONI TRIG. - Listing. Lo scaling in direzione X è fisso, può invece variare quello in direzione Y, variando il valore SF di riga 120, che indica il DY in coordinate schermo, da far corrispondere all'unità.

```
10  REM  DISEGNO DI UNA FAMIGLIA DI SINUSOIDI
20 PI = 3.1416:X1 = PI * 3:SX = PI / 8:SS = 132.63:FL = 1
30 D$ =  CHR$ (4):A$ = "DISEGNO DI UNA FAMIGLIA DI SINUSOIDI"
40  HGR : HCOLOR= 3: HOME
50 XS = 0:YS = 1600: GOSUB 300: GOSUB 320
60 YS = 0: GOSUB 300:XS = 2500: GOSUB 310:YS = 1580
70  GOSUB 310:XS = 0: GOSUB 310:YS = 0: GOSUB 310
80  FOR XS = 208 TO 2500 STEP 416:YS = 0
90  GOSUB 300:YS = 1580: GOSUB 310: NEXT XS
100  FOR SY = 1300 TO 0 STEP  - 100
110  FOR X =  - X1 TO X1 + .001 STEP SX:Y =  SIN (X) * (SY / 5)
120 XS = 1250.5 + X * SS:YS = Y + SY + .5
130  IF X =  - X1 THEN  GOSUB 300
140  GOSUB 310: NEXT X,SY: END
300  IF FL = 1 THEN  HPLOT XS / 10,YS / 10: RETURN
301  PRINT D$"PR#1": PRINT "M";XS;",";YS: PRINT D$"PR#0": RETURN
310  IF FL = 1 THEN  HPLOT  TO XS / 10,YS / 10: RETURN
311  PRINT D$"PR#1": PRINT "D";XS;",";YS: PRINT D$"PR#0": RETURN
320  IF FL = 1 THEN  VTAB (23): PRINT A$: RETURN
321  PRINT D$"PR#1": PRINT "P";A$: PRINT D$" PR# 0": RETURN
```

Figura 6 - PROGRAMMA SINUSOIDI - Listing. Questo, come i programmi seguenti, ha una uscita su monitor, ponendo il flag F. = 1. Ponendo invece il FL = O, esce sul plotter Watanabe, il cui software è stato descritto negli scorsi numeri.

e ogni punto P è individuato da un raggio R (distanza tra il punto e l'origine) e da un angolo (formato tra il segmento PO e l'asse di riferimento).

Esistono delle semplici formule di passaggio tra le coordinate cartesiane P (X,Y) e le coordinate polari P (R, α) (Figura 11).

Mentre però il passaggio tra coordinate polari e coordinate cartesiane è univoco non è vero il viceversa. Quindi per la funzione arcotangente, che può essere definita solo a meno di 180°, vanno inseriti dei controlli per determinare il quadrante di appartenenza dell'angolo.

Per parecchie figure geometriche quindi esistono formule in coordinate cartesiane e in coordinate polari.

Ad esempio il cerchio con centro nell'origine degli assi espresso in coordinate cartesiane
Y = SQR (R^2 - X^2)
e espresso in coordinate polari
X = R* COS (α); Y = R* SIN (α)
Con α che varia da Ø a 2* PIgreco
Da un punto di vista computer grafico è spesso meglio utilizzare, per la rappresentazione di linee curve, coordinate polari, ovviamente quando questo sia possibile.

Il programma ellisse

Anche l'ellisse può essere espressa in coordinate polari, prendendo come origine del riferimento un fuoco e come asse polare quello relativo all'asse maggiore. La formula risolutiva, riportata in molti testi è simile a quella della circonferenza
X = R* COS (α)
Y = R* SIN (α)
con la differenza che il Raggio R varia anche esso col variare dell'angolo R = P/(1 + E* COS (α))
dove P, ordinata dal fuoco, e E eccentricità sono grandezze caratteristiche dell'ellisse, determinabili tramite i dati, forniti da input, relativi ad asse maggiore ed asse minore.

Il programma presentato (list. figura 12 e output figura 13) gira anch'esso su moni-



Figura 5 - PROGRAMMA GRAFICO FUNZIONI TRIG. - Output. Nello stesso output sono riportate curve il cui valore massimo è 1 e curve il cui valore massimo è INFINITO. È ovvio che queste ultime "escono fuori".



Figura 7 - PROGRAMMA SINUSOIDI - Output. È evidentemente un programma "decorativo". Non ha infatti nessun significato sovrapporre sinusoidi, variando dei coefficienti.

$$R = SQR(X\hat{\ }2 + Y\hat{\ }2)$$

$$A = ATN(X/Y)$$



Figura 11 - COORDINATE POLARI E COORDINATE CARTESIANE - Quando nei programmi grafici viene rappresentata la rotazione di punti o di figure rispetto a punti, è indispensabile lavorare in coordinate polari.

```
100  REM  DISEGNO DI UNA CICLOIDE SU PLOTTER
110 R = 150:PI = 3.1415926:PP = PI / 20:FL = 1
120 D$ =  CHR$ (4):X$ = "DISEGNO DI UNA FAMIGLIA DI CICLOIDI"
130  HGR : HCOLOR= 3: HOME :X = 0:Y = 790: GOSUB 300
140 X = 2600:Y = 790: GOSUB 310
150  FOR B = 0 TO 3 * R STEP R / 3
160  FOR A =  - 2.5 * PI TO 2.5 * PI STEP PP
170 X =  INT (R * A - B *  SIN (A) + 1300)
180 Y =  INT (R - B *  COS (A) + 790)
190  IF A =  - 2.5 * PI THEN  GOSUB 300
200  GOSUB 310: NEXT A, B
210 X = 0:Y = 200: GOSUB 300: GOSUB 320: END
300  IF FL = 1 THEN  HPLOT X / 10, Y / 10: RETURN
301  PRINT D$"PR#1": PRINT "M";X;",";Y: PRINT D$"PR#0": RETURN
310  IF FL = 1 THEN  HPLOT  TO X / 10, Y / 10: RETURN
311  PRINT D$"PR#1": PRINT "D";X;",";Y: PRINT D$"PR#0": RETURN
320  IF FL = 1 THEN  VTAB (23): PRINT X$: RETURN
321  PRINT D$"PR#1": PRINT "P";X$: PRINT D$"PR#0": RETURN
```

Figura 8 - PROGRAMMA CICLOIDE - Listing. Nei programmi con output solo grafico i loop di calcolo delle funzioni trigonometriche si fanno direttamente in radianti.



Figura 9-10 - PROGRAMMA CICLOIDE - Output. I due output dello stesso programma si differenziano solo per lo step pari a PI/20, cioè 9° nella figura 9 e a PI/2 cioè 90° nella figura 10.

```
10  HOME : PRINT "  DISEGNO DI UNA ELLISSE"
20  PRINT "  IN COORDINATE POLARI "
30  VTAB (7): INPUT " ASSE MAGGIORE              ";A
40  VTAB (9): INPUT " ASSE MINORE                ";B
50 C =  SQR (A ^ 2 - B ^ 2):E = C / A
60  VTAB (12): PRINT " ECCENTRICITA'              ";E
70 P = A * (1 - E ^ 2)
80  VTAB (14): PRINT " ORDINATA DI UN FUOCO    ";P
90 L = 3.14159 * (3 * (A + B) / 2 -  SQR (A * B))
100  VTAB (17): PRINT " LUNGHEZZA                 ";L
110 S = 3.14159 * A * B:D$ =  CHR$ (4)
120  VTAB (19): PRINT " AREA                      ";S
130  VTAB (23): INPUT "PREMI RETURN PER CONTINUARE ";R$
140 FL = 1: HGR : HCOLOR= 3: GOSUB 300
170 X = 2400: GOSUB 310:Y = 1580: GOSUB 310:X = 0:Y = 1580
180  GOSUB 310:Y = 0: GOSUB 310:Y = 790: GOSUB 300:X = 2400
190  GOSUB 310:X = 1200:Y = 0: GOSUB 300:Y = 1580: GOSUB 310
200  FOR A = 0 TO 6.3 STEP .1
210 X = 1200.5 + C:Y = 790.5: GOSUB 300
220 R = P / (1 + E *  COS (A))
230 X =  INT (1200.5 + C + R *  COS (A))
240 Y =  INT (790.5 + R *  SIN (A))
250  GOSUB 310: NEXT A
270 X$ = "DISEGNO DI UNA ELLISSE": PRINT "IN COORDINATE POLARI"
280 X = 0:Y = 1650: GOSUB 300: GOSUB 320: END
300  IF FL = 1 THEN  HPLOT X / 10,Y / 10: RETURN
301  PRINT D$"PR#1": PRINT "M";X;",";Y: PRINT D$"PR#0": RETURN
310  IF FL = 1 THEN  HPLOT  TO X / 10,Y / 10: RETURN
311  PRINT D$"PR#1": PRINT "D";X;",";Y: PRINT D$"PR#0": RETURN
320  IF FL = 1 THEN  VTAB (23): PRINT X$: RETURN
321  PRINT D$"PR#1": PRINT "P";X$: PRINT D$" PR# 0": RETURN
```

Figura 12 - PROGRAMMA ELLISSE - Listing. La formula da usare in coordinate cartesiane: Y = ± (B/A) SQR (A^2 - X^2). Abbiamo invece utilizzato quella in coordinate polari, con il polo coincidente con un fuoco dell'ellisse e con l'asse polare coincidente con l'asse X.*

tor e plotter, visualizza dapprima tutti i valori caratteristici, come ordinata del fuoco rispetto al riferimento cartesiano, la eccentricità, (pari a 1 nel caso del cerchio), lunghezza della curva e superficie dell'area compresa.

Viene poi visualizzata l'ellisse tramite una raggiera uscente dall'origine del riferimento polare.

Il programma spirale di Archimede

Un'altra curva nota anche ai non addetti ai lavori è la spirale.

Ne esistono di vari tipi che si sviluppano o sul piano o nello spazio.

La più semplice è la spirale di Archimede.

Immaginate di tracciare una circonferenza con un compasso, ma mentre tracciate la linea il compasso si apre sempre di più. La linea non si chiude e diventa una spirale di Archimede. Il programma realizzato (listato in figura 14 e output su plotter in figura 15) utilizza ancora le coordinate polari

X = R* COS (A)
Y = R* SIN (A)

Solo che come abbiamo detto anche il Raggio varia linearmente con l'angolo A (riga 160).



Figura 13 - PROGRAMMA ELLISSE - Output. Lo step del loop sull'angolo è costante, ma poichè i raggi della raggiera sono di lunghezza variabile, il disegno è asimmetrico.

```
90  REM  SPIRALE DI ARCHIMEDE
100 FL = 1: HGR : HCOLOR= 3: HOME
110 PI = 3.14159:SP = PI / 50:R0 = 300:D$ =  CHR$ (4): GOSUB 300
120 X = 2600: GOSUB 310:Y = 1580: GOSUB 310:X = 0:Y = 1580
130  GOSUB 310:Y = 0: GOSUB 310:Y = 790: GOSUB 300:X = 2600
140  GOSUB 310:X = 1300:Y = 0: GOSUB 300:Y = 1580: GOSUB 310
150  FOR A = 0 TO 100 * PI STEP SP
160 R = R0 * A / (PI * 2)
170 X =  INT (1300.5 + R *  SIN (A))
180 Y =  INT (790.5 + R *  COS (A))
190  IF X < 0 THEN X = 0
200  IF Y < 0 THEN Y = 0
210  IF X > 2600 THEN X = 2600
220  IF Y > 1580 THEN Y = 1580
230  IF R > 1520 THEN 260
240  IF A = 0 THEN  GOSUB 300: NEXT A
250  GOSUB 310: NEXT A
260 X$ = "SPIRALE DI ARCHIMEDE"
270 X = 0:Y = 1580: GOSUB 300: GOSUB 320: END
300  IF FL = 1 THEN  HPLOT X / 10,Y / 10: RETURN
301  PRINT D$"PR#1": PRINT "M";X;",";Y: PRINT D$"PR#0": RETURN
310  IF FL = 1 THEN  HPLOT  TO X / 10,Y / 10: RETURN
311  PRINT D$"PR#1": PRINT "D";X;",";Y: PRINT D$"PR#0": RETURN
320  IF FL = 1 THEN  VTAB (23): PRINT X$: RETURN
321  PRINT D$"PR#1": PRINT "P";X$: PRINT D$"PR#0": RETURN
```

Figura 14 - PROGRAMMA SPIRALE DI ARCHIMEDE - Listing. La formula della spirale è come quella del cerchio solo che il raggio aumenta con l'aumentare dell'angolo e quindi la curva non si chiude mai.



Figura 15 - PROGRAMMA SPIRALE DI ARCHIMEDE - Output. Il programma ha controlli di formato, per evitare la fuoriuscita del disegno dal bordo e controllo di fine, che interrompe l'esecuzione quando il raggio supera la dimensione della semidiagonale del formato di uscita.

Vi sono dei controlli di formato per evitare che la curva esca dalla cornice e un controllo di fine. Il programma finisce quando il Raggio, che aumenta via via, diventa più grande della semi-diagonale del rettangolo di uscita, e quindi la curva ormai è tutta esterna.

In questo articolo abbiamo esaminato quindi le funzioni trigonometriche e le abbiamo utilizzate in vari programmi grafici dimostrativi.

Abbiamo utilizzato anche il plotter per l'output dei programmi, in alternativa al monitor.

Ricordiamo che il problema dell'output su plotter è stato esaminato sul numero 5 della rivista.

Nel prossimo numero continueremo a trattare di trigonometria, anche dal punto di vista soluzione dei triangoli.

Maurizio Petroni

PERIFERICHE PER TUTTI

* TASTIERA ALFANUMERICA PROFESSIONALE



77 tasti con pad numerico e funzioni
Full ASCII - cinque funzioni — L. 175.000
In contenitore plastico — L. 245.000

* TERMINALE INTERATTIVO

Monitor 12" - Tastiera da 82 tasti.
Display 80 x 24; 1920 caratteri - 2 Pagine
Linea di status, highlighting, funzioni speciali — L. 985.000

* STAMPANTI

Controllo a microprocessore - Interfaccia parallela
Percorso bidirezionale ottimizzato

L. 11	80 col.	100 cps	L. 700.000
L. 31	132 col.	100 cps	L. 850.000
L. 26	132 col.	160 cps	L. 2.100.000

* DISK DRIVES



Drive 5" doppia faccia - doppia densità	(500 Kbytes)	L. 387.000
Drive 8" doppia faccia - doppia densità	(1.6 Mbytes)	L. 650.000
Drive 5" hard disk	(7.5 Mbytes)	L. 1.780.000

* MONITOR PROFESSIONALE 12"



Input video: 1 Vpp - 75 Ohm
Banda video: 10 hz ± 24 Mhz a 3 db
Fosfori verdi P31
Completo di alimentazione e cavo di rete — L. 185.000

* CONTROLLERS

- Video controller
- Graphic processor
- Floppy disk controller
- Hard disk controller
- Schede a microprocessore per usi industriali.

Tutti i prodotti sono garantiti dalla KYBER, azienda italiana leader nella produzione di sistemi di elaborazione.

Prezzi così competitivi (non legati al dollaro) sono resi possibili grazie alla grande movimentazione delle quantità determinate dalla produzione KYBER

SCONTI PER QUANTITÀ

KYBER ®
CALCOLATORI

via Bellaria 54-58 - 51100 PISTOIA - Tel. 0573/368113 (2 linee)